A FriCAS Package for Combinatorial Probability Theory

Franz Lehner

Preliminary version, April 9, 2021

Abstract

We present a package for combinatorial manipulation of probability distributions, moments, cumulants, orthogonal polynomials and convolutions. The central object is the FriCAS domain Distribution, which represents probability distributions as streams of moments and cumulants.

Contents

1 A few words about Axiom/FriCAS

1.1 What is FriCAS?

FriCAS is a CAS (computer algebra system) from the AXIOM family and can do most things CAS usually do, namely exact computations with numbers and formal expressions

1a ⟨*example1.input* 1a⟩≡ 1b ▷

```
p := 1+x+x^2
```

$$x^2 + x + 1 \tag{1}$$

```
Type: Polynomial(Integer)
```

like differentiation

1b ⟨*example1.input* 1a⟩+≡ ◁ 1a 1c ▷

```
D(p, x)
```

$$2\,x + 1 \tag{2}$$

```
Type: Polynomial(Integer)
```

and integration (it has the most complete implementation of the Risch algorithm):

1c ⟨*example1.input* 1a⟩+≡ ◁ 1b 2a ▷

```
integrate(p, x)
```

$$\frac{1}{3}\,x^3+\frac{1}{2}\,x^2+x \tag{3}$$

Type: Polynomial(Fraction(Integer))

1.2 Types and Domains

The main difference to other CAS is *strong typing*, i.e., every object has a type, called *Domain*, which usually is some set of mathematical objects. This is very handy when it comes to computing with non-standard things like finite fields or noncommutative rings:

2a ⟨*example1.input* 1a⟩+≡ ◁1c 2b▷

```
FG := FreeGroup Symbol
a:FG := 'a
b:FG := 'b

a*b*a^-1 *b^-1
```

$$a\ b\ a^{(-1)}\ b^{(-1)} \tag{4}$$

Type: FreeGroup(Symbol)

If the desired type is not indicated, the interpreter will guess one.

Domains can be built on top of each other:

2b ⟨*example1.input* 1a⟩+≡ ◁2a 2c▷

```
FGA := MonoidRing (Integer, FG)
a1 := a::FGA
a1i := (a^-1)::FGA
b1 := b::FGA
b1i := (b^-1)::FGA
T := a1+a1i+b1+b1i
```

$$b^{(-1)}+a^{(-1)}+a+b \tag{5}$$

Type: MonoidRing(Integer, FreeGroup(Symbol))

Through the type system, FriCAS knows which multiplication it is meant to perform:

2c ⟨*example1.input* 1a⟩+≡ ◁2b 2d▷

```
T^2
```

$$b^2+b\ a+b\ a^{(-1)}+a\ b+a^2+a\ b^{(-1)}+a^{(-1)}\ b+a^{(-2)}+a^{(-1)}\ b^{(-1)}+4+b^{(-1)}\ a+b^{(-1)}\ a^{(-1)}+b^{(-2)} \tag{6}$$

Type: MonoidRing(Integer, FreeGroup(Symbol))

2d ⟨*example1.input* 1a⟩+≡ ◁2c 2e▷

```
A:Matrix FGA := matrix [[a1, a1i], [b1, b1i]]
```

$$\begin{bmatrix} a & a^{(-1)} \\ b & b^{(-1)} \end{bmatrix} \tag{7}$$

Type: Matrix(MonoidRing(Integer, FreeGroup(Symbol)))

2e ⟨*example1.input* 1a⟩+≡ ◁2d 2f▷

```
A*A
```

$$\begin{bmatrix} a^{(-1)}\ b+a^2 & a^{(-1)}\ b^{(-1)}+1 \\ 1+b\ a & b^{(-2)}+b\ a^{(-1)} \end{bmatrix} \tag{8}$$

Type: Matrix(MonoidRing(Integer, FreeGroup(Symbol)))

Types can be built on top of each other only if it makes sense:

2f ⟨*example1.input* 1a⟩+≡ ◁2e 3a▷

```
B:Matrix String := matrix [["a", "b"], ["c", "d"]]
```

```
Matrix(String) is not a valid type.
```

See the Axiom book [1] by Jenks and Sutor for more details.

1.3 Streams

The axiom clones (Axiom/FriCAS/OpenAxiom) are the only CAS supporting streams, i.e., lists of indefinite length, which allow to represent infinite mathematical objects (more precisely, potentially infinite objects in the sense of Aristotle), and delayed computation (things are computed when needed). This is very convenient to compute with infinite sequences like moments and cumulants.

3a ⟨*example1.input* 1a⟩+≡ ◁ 2f 3b ▷

```
nat := [n for n in 1..]
```

$$[1,\ 2,\ 3,\ 4,\ 5,\ 6,\ 7,\ 8,\ 9,\ 10,\ ...] \tag{9}$$

```
Type: Stream(PositiveInteger)
```

3b ⟨*example1.input* 1a⟩+≡ ◁ 3a 3c ▷

```
prim := [n for n in nat | prime? n]
```

$$[2,\ 3,\ 5,\ 7,\ 11,\ 13,\ 17,\ 19,\ 23,\ 29,\ ...] \tag{10}$$

```
Type: Stream(PositiveInteger)
```

Streams are used for the representation of power series, this way we don't have to worry about the number of terms we want to compute

3c ⟨*example1.input* 1a⟩+≡ ◁ 3b 3d ▷

```
exps := series(exp x, x=0)
```

$$1+x+\frac{1}{2}\ x^2+\frac{1}{6}\ x^3+\frac{1}{24}\ x^4+\frac{1}{120}\ x^5+\frac{1}{720}\ x^6+\frac{1}{5040}\ x^7+\frac{1}{40320}\ x^8+\frac{1}{362880}\ x^9+\frac{1}{3628800}\ x^{10}+O\left(x^{11}\right) \tag{11}$$

```
Type: UnivariatePuiseuxSeries(Expression(Integer), x, 0)
```

The number of terms computed by default is controlled with the command

3d ⟨*example1.input* 1a⟩+≡ ◁ 3c 3e ▷

```
)set stream calculate 5
exps := series(exp x, x=0)
```

$$1+x+\frac{1}{2}\ x^2+\frac{1}{6}\ x^3+\frac{1}{24}\ x^4+\frac{1}{120}\ x^5+O\left(x^6\right) \tag{12}$$

```
Type: UnivariatePuiseuxSeries(Expression(Integer), x, 0)
```

Yet we can compute as far as we want

3e ⟨*example1.input* 1a⟩+≡ ◁ 3d

```
coefficient(exps, 50)
```

$$\frac{1}{30414093201713378043612608166064768844377641568960512000000000000} \tag{13}$$

```
Type: Expression(Integer)
```

2 The Distribution package

2.1 Implementation

`FriCAS` can be extended by the user with her own data types. In our case, a probability distribution is represented by its sequence of moments

$$[m_1, m_2, \dots]$$

over a ring. The moments and the various cumulants (classical, free, boolean) are stored as streams. Distributions can be constructed from a given stream of moments or cumulants, the remaining cumulants are computed automatically on demand. In addition, new distributions can be constructed using various convolution operations, like classical, free, boolean and monotone additive convolutions, classical and free multiplicative convolutions, and orthogonal and subordination convolutions. Moreover, orthogonal polynomials and Jacobi parameters can be computed.

You can see all available commands by typing

```
)sh Distribution
```

which currently shows the following:

```
)sh DISTRO
 Distribution(R: CommutativeRing) is a domain constructor
 Abbreviation for Distribution is DISTRO
 This constructor is exposed in this frame.
------------------------------ Operations ------------------------------

 ?=? : (%, %) -> Boolean                0 : () -> %
 ?^? : (%, PositiveInteger) -> %        coerce : % -> OutputForm
 freeConvolution : (%, %) -> %          freeCumulants : % -> Sequence(R)
 hash : % -> SingleInteger              latex : % -> String
 moments : % -> Sequence(R)             ?~=? : (%, %) -> Boolean
 booleanConvolution : (%, %) -> %
 booleanCumulant : (%, PositiveInteger) -> R
 booleanCumulantFromJacobi : (Integer, Sequence(R), Sequence(R)) -> R
 booleanCumulants : % -> Sequence(R)
 classicalConvolution : (%, %) -> %
 classicalCumulant : (%, PositiveInteger) -> R
 classicalCumulants : % -> Sequence(R)
 construct : (Sequence(R), Sequence(R), Sequence(R), Sequence(R)) -> %
 distributionByBooleanCumulants : Stream(R) -> %
 distributionByBooleanCumulants : Sequence(R) -> %
 distributionByClassicalCumulants : Stream(R) -> %
 distributionByClassicalCumulants : Sequence(R) -> %
 distributionByEvenMoments : Stream(R) -> %
 distributionByEvenMoments : Sequence(R) -> %
 distributionByFreeCumulants : Stream(R) -> %
 distributionByFreeCumulants : Sequence(R) -> %
 distributionByJacobiParameters : (Stream(R), Stream(R)) -> %
 distributionByJacobiParameters : (Sequence(R), Sequence(R)) -> %
 distributionByMoments : Stream(R) -> %
 distributionByMoments : Sequence(R) -> %
 distributionByMonotoneCumulants : Stream(R) -> % if R has ALGEBRA(FRAC(INT))
 distributionByMonotoneCumulants : Sequence(R) -> % if R has ALGEBRA(FRAC(INT))
 distributionBySTransform : (Fraction(Integer), Fraction(Integer), Sequence(R)) -> % if
 distributionBySTransform : Record(puiseux: Fraction(Integer),laurent: Fraction(Integer)
 freeCumulant : (%, PositiveInteger) -> R
 freeMultiplicativeConvolution : (%, %) -> % if R has ALGEBRA(FRAC(INT))
 hankelDeterminants : % -> Stream(R)
```

```
hashUpdate! : (HashState, %) -> HashState
jacobiParameters : % -> Record(an: Stream(Fraction(R)),bn: Stream(Fraction(R))) if R ha
jacobiParameters : % -> Record(an: Stream(R),bn: Stream(R)) if R has FIELD
moment : (%, NonNegativeInteger) -> R
monotoneConvolution : (%, %) -> %
monotoneCumulants : % -> Sequence(R) if R has ALGEBRA(FRAC(INT))
orthogonalConvolution : (%, %) -> %
orthogonalPolynomials : % -> Stream(SparseUnivariatePolynomial(Fraction(R))) if R has I
orthogonalPolynomials : % -> Stream(SparseUnivariatePolynomial(R)) if R has FIELD
subordinationConvolution : (%, %) -> %
```

Other commands are in different packages

```
)sh STransformPackage
)sh DistributionPolynomialPackage
)sh DistributionContinuedFractionPackage
)sh DISTEX DistributionPackage
```

All parameters are modeled as `Streams`, which has the following advantages:

- Calculation of moments and cumulants is delayed until demand
- once calculated, they are cached.
- the user does not have to worry about a maximal order of moments beforehand, the system computes as far as time and RAM restrictions of the underlying LISP and hardware allow.

Once an interesting sequence is found, the `GUESS` package by M. Rubey (`arXiv:math/0702086`, also part of `FriCAS`) can be used to guess a recurrence or closed formula.

2.2 Example: Creating distributions via cumulant and moment sequences

Say we want to check whether the normal distribution is infinite divisible with respect to free convolution. There are several ways to construct the normal distribution.

1. The quickest way to get the normal distribution is via its classical cumulants, because only the second cumulant is nonzero. We need a declaration of an integer stream, because by default a nonnegative stream is generated and nonnegative integers do not form a ring.

5a ⟨*example2.input* 5a⟩≡ 5b ▷

```
gausscum:Stream Integer := concat([0, 1], repeating [0])
```

$$\left[0,\ 1,\ \overline{0}\right] \tag{14}$$

```
                                                    Type:  Stream(Integer)
```

5b ⟨*example2.input* 5a⟩+≡ ◁ 5a 6a ▷

```
gauss := distributionByClassicalCumulants gausscum
```

$$[0,\ 1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ ...] \tag{15}$$

Type: Distribution(Integer)

2. By the explicit formula for its moments $m_{2k} = \frac{2k!}{2^k k!}$:

6a ⟨*example2.input* 5a⟩+≡ ◁5b 6b▷

```
gaussmom k ==
   odd? k => return 0
   k2 := exquo(k, 2)
   return (factorial(k)/(2^k2*factorial k2))::Integer

gauss := distributionByMoments [gaussmom k for k in 1..]
```

$$[0,\ 1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ ...] \tag{16}$$

Type: Distribution(Integer)

3. By dissecting the Taylor series of the exponential moment generating function $e^{z^2/2}$:

6b ⟨*example2.input* 5a⟩+≡ ◁6a 6c▷

```
gaussexp := series(exp(z^2/2), z=0)
```

$$1 + \frac{1}{2}\ z^2 + \frac{1}{8}\ z^4 + \frac{1}{48}\ z^6 + \frac{1}{384}\ z^8 + \frac{1}{3840}\ z^{10} + O\left(z^{11}\right) \tag{17}$$

Type: UnivariatePuiseuxSeries(Expression(Integer), z, 0)

Next we must extract the coefficients of this series

6c ⟨*example2.input* 5a⟩+≡ ◁6b 6d▷

```
coefficients gaussexp
```

$$\left[1,\ 0,\ \frac{1}{2},\ 0,\ \frac{1}{8},\ 0,\ \frac{1}{48},\ 0,\ \frac{1}{384},\ 0,\ ...\right] \tag{18}$$

Type: Stream(Expression(Integer))

and multiply term by term with $n!$:

6d ⟨*example2.input* 5a⟩+≡ ◁6c 6e▷

```
gaussmom := [c*factorial n  for n in 1.. for c in coefficients gaussexp]
```

$$[1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ 0,\ ...] \tag{19}$$

Type: Stream(Expression(Integer))

4. As moments of the position operator in the Weyl algebra of differential operators, which is already implemented in FriCAS.

6e ⟨*example2.input* 5a⟩+≡ ◁6d 6f▷

```
R := UP(z, Integer)
W := LODO1 R
```

LinearOrdinaryDifferentialOperator1(Fraction(UnivariatePolynomial(z, Integer))) (20)

Now z and d verify the canonical commutation relation:

6f ⟨*example2.input* 5a⟩+≡ ◁6e 7a▷

```
z:R := 'z
d:W := D()
d*z-z*d
```

$$1 \tag{21}$$

Type: LinearOrdinaryDifferentialOperator1(Fraction(UnivariatePolynomial(z,Integer)))

we can compute the moments in this algebra by extracting the constant coefficient:

7a ⟨*example2.input* 5a⟩+≡ ◁6f 7b▷

```
g6 := (d+z)^6
```

$$D^6+6\ z\ D^5+\left(15\ z^2+15\right)\ D^4+\left(20\ z^3+60\ z\right)\ D^3+\left(15\ z^4+90\ z^2+45\right)\ D^2+\left(6\ z^5+60\ z^3+90\ z\right)\ L \tag{22}$$

Type: LinearOrdinaryDifferentialOperator1(UnivariatePolynomial(z,Integer))

The Weyl algebra is generated by D with coefficient ring $\mathbf{C}[z]$:

7b ⟨*example2.input* 5a⟩+≡ ◁7a 7c▷

```
coefficient(g6, 0)
```

$$z^6+15\ z^4+45\ z^2+15 \tag{23}$$

Type: UnivariatePolynomial(z,Integer)

Therefore we must extract the constant coefficient of this polynomial:

7c ⟨*example2.input* 5a⟩+≡ ◁7b 7d▷

```
coefficient(coefficient(g6, 0), 0)
```

and finally we obtain the gaussian moments

7d ⟨*example2.input* 5a⟩+≡ ◁7c 7e▷

```
g := distributionByMoments([coefficient(coefficient((d+z)^k, 0), 0) for k in 1..]
```

$$[0,\ 1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ ...] \tag{24}$$

Type: Distribution(Integer)

5. By its Jacobi parameters:

7e ⟨*example2.input* 5a⟩+≡ ◁7d 7f▷

```
an:Stream Integer := repeating [0]
```

$$[\overline{0}] \tag{25}$$

Type: Stream(Integer)

7f ⟨*example2.input* 5a⟩+≡ ◁7e 7g▷

```
bn:Stream Integer := [k for k in 1..]
```

$$[1,\ 2,\ 3,\ 4,\ 5,\ 6,\ 7,\ 8,\ 9,\ 10,\ ...] \tag{26}$$

Type: Stream(Integer)

7g ⟨*example2.input* 5a⟩+≡ ◁7f 7h▷

```
gauss := distributionByJacobiParameters(an, bn)
```

$$[0,\ 1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ ...] \tag{27}$$

Type: Distribution(Integer)

6. In the case of the normal distribution we can also use the built-in function

7h ⟨*example2.input* 5a⟩+≡ ◁7g 8b▷

```
gauss := gaussianDistribution 1
```

$$[0,\ 1,\ 0,\ 3,\ 0,\ 15,\ 0,\ 105,\ 0,\ 945,\ ...] \tag{28}$$

Type: Distribution(Integer)

Infinite divisibility means that the sequence of free cumulants,

8a ⟨*example.input* 8a⟩≡

```
freeCumulants gauss
```

$$[0,\ 1,\ 0,\ 1,\ 0,\ 4,\ 0,\ 27,\ 0,\ 248,\ ...] \tag{29}$$

Type: Sequence(Integer)

starting with the third element, is positive definite and thus forms the moment sequence of the Levy measure:

8b ⟨*example2.input* 5a⟩+≡ ◁ 7h 8c ▷

```
levymom := rest rest freeCumulants gauss;
levy := distributionByMoments levymom
```

$$[0,\ 1,\ 0,\ 4,\ 0,\ 27,\ 0,\ 248,\ 0,\ 2830,\ ...] \tag{30}$$

Type: Distribution(Integer)

Now we need to check whether the Hankel determinants are nonnegative:

8c ⟨*example2.input* 5a⟩+≡ ◁ 8b 8d ▷

```
hankelDeterminants(levy)
```

[1, 3, 33, 1837, 586337, 1319746279, 23073023457505, 3650976798436751385, 565487262774358704480003

(31)

Type: Stream(Integer)

Equivalently, we can check the positivity of the Jacobi parameters:

8d ⟨*example2.input* 5a⟩+≡ ◁ 8c

```
jacobiParameters levy
```

$$[an = [0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ 0,\ ...],$$
$$bn = \left[1,\ 3,\ \frac{11}{3},\ \frac{167}{33},\ \frac{10533}{1837},\ \frac{4134779}{586337},\ \frac{10250885015}{1319746279},\ \frac{208831020924783}{23073023457505},\ \frac{11912337984453599843}{1216992266145583795},\ \frac{694320}{62831}\right.$$

Type: Record(an: Stream(Fraction(Integer)),bn: Stream(Fraction(Integer)))

2.3 Example: Convolutions

Computing convolutions is easy, using the commands `classicalConvolution`, `freeConvolution`, `freeMultiplicativeConvolution`,...

We investigate the behaviour of Jacobi Parameters under free convolution with a Wigner distribution.

8e ⟨*example3.input* 8e⟩≡ 8f ▷

```
w := wignerDistribution t
```

$$\left[0,\ t,\ 0,\ 2\ t^2,\ 0,\ 5\ t^3,\ 0,\ 14\ t^4,\ 0,\ 42\ t^5,\ ...\right] \tag{33}$$

Type: Distribution(Polynomial(Integer))

8f ⟨*example3.input* 8e⟩+≡ ◁ 8e 9a ▷

```
x := distributionByJacobiParameters([a[k] for k in 0..], [b[k] for k in 0..])
```

$$\left[a_0,\ b_0 + {a_0}^2,\ (a_1 + 2\ a_0)\ b_0 + {a_0}^3,\ ...\right] \tag{34}$$

Type: Distribution(Polynomial(Integer))

9a ⟨*example3.input* 8e⟩+≡ ◁8f 9b▷

```
xw := freeConvolution(x,w)
```

$$\left[a_0,\ t + b_0 + {a_0}^2,\ 3\ a_0\ t + (a_1 + 2\ a_0)\ b_0 + {a_0}^3,\ ...\right] \tag{35}$$

Type: Distribution(Polynomial(Integer))

9b ⟨*example3.input* 8e⟩+≡ ◁9a ?? ▷

```
jacobiParameters xw
```

$$an = \left[a_0,\ \frac{a_0\ t + a_1\ b_0}{t + b_0},\ \frac{a_0\ t^4 + (a_1 + 3\ a_0)\ b_0\ t^3 + \left((a_2 + 2\ a_1 - 2\ a_0)\ b_0\ b_1 + 5\ a_0\ {b_0}^2 + ({a_1}^3 - 2\ a_0\ {a_1}^2 + \right.}{t^4 + 4\ b_0\ t^3 + \left(b_0\ b_1 + 5\ {b_0}^2 + ({a_1}^2 - 2\ a_0\ a_1\right.}\right.$$

$$bn = \left[t + b_0,\ \frac{t^3 + 3\ b_0\ t^2 + \left(b_0\ b_1 + 2\ {b_0}^2 + ({a_1}^2 - 2\ a_0\ a_1 + {a_0}^2)\ b_0\right)\ t + {b_0}^2\ b_1}{t^2 + 2\ b_0\ t + {b_0}^2},\ \frac{t^7 + 7\ b_0\ t^6 + \left(3\ b_0\ b_1 + 1\right.}{}\right. \tag{36}$$

Type: Record(an: Stream(Fraction(Polynomial(Integer))),bn: Stream(Fraction(Polynomial(Integer))))

2.4 Example: Orthogonal polynomials

We verify that up to a rescaling the orthogonal polynomials of the Wigner distribution are indeed the Chebyshev polynomials, which can be obtained with the command `chebyshevU`:

9c ⟨*example4.input* 9c⟩≡ 9d▷

```
ch2 := orthogonalPolynomials(w)
```

$$\left[?,\ ?^2 - 1,\ ?^3 - 2\ ?,\ ?^4 - 3\ ?^2 + 1,\ ?^5 - 4\ ?^3 + 3\ ?,\ ...\right] \tag{37}$$

Type: Stream(SparseUnivariatePolynomial(Fraction(Integer)))

To give the unknown variable a name, convert it to the polynomial domain $\mathbf{Z}[t]$:

9d ⟨*example4.input* 9c⟩+≡ ◁9c 9e▷

```
ch2::Stream UP(t, Integer)
```

$$\left[t,\ t^2 - 1,\ t^3 - 2\ t,\ t^4 - 3\ t^2 + 1,\ t^5 - 4\ t^3 + 3\ t,\ ...\right] \tag{38}$$

Type: Stream(UnivariatePolynomial(t,Integer))

9e ⟨*example4.input* 9c⟩+≡ ◁9d 9f▷

```
chu := [chebyshevU(k,t/2) for k in 1..]
```

$$\left[t,\ t^2 - 1,\ t^3 - 2\ t,\ t^4 - 3\ t^2 + 1,\ t^5 - 4\ t^3 + 3\ t,\ ...\right] \tag{39}$$

Type: Stream(Polynomial(Fraction(Integer)))

Now we compare the first ten polynomials:

9f ⟨*example4.input* 9c⟩+≡ ◁9e

```
ch2a := ch2::Stream UP(t, Integer)
chua := chu::Stream UP(t, Integer)
(entries complete first(ch2a, 10) = entries complete first(chua, 10))::Boolean
```

$$true \tag{40}$$

Type: Boolean

2.5 Further examples

We verify the convolution identity [2, p.2, (6)]

$$\mu \rhd \nu = \nu \uplus (\mu \vdash \nu) \tag{41}$$

10a ⟨*example5.input* 10a⟩≡ ?? ▷

```
mu:=distributionByMoments([a[k] for k in 1..])
nu:=distributionByMoments([b[k] for k in 1..])
mumonnu := monotoneConvolution(mu, nu);
mutestnu:= booleanConvolution(nu, orthogonalConvolution(mu, nu));
moments mumonnu - moments mutestnu
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, ...] \tag{42}$$

Next we verify the convolution identity defining subordination convolution

$$\mu \boxplus \nu = \nu \rhd (\mu \boxplus \nu)$$

To this end we express both distributions in terms of free cumulants

10b ⟨*example6.input* 10b⟩≡ ?? ▷

```
mu := distributionByFreeCumulants([a[k] for k in 1..])
nu := distributionByFreeCumulants([b[k] for k in 1..])
freeCumulants monotoneConvolution(nu, subordinationConvolution(mu, nu))
```

$$[b_1 + a_1,\ b_2 + a_2,\ b_3 + a_3,\ b_4 + a_4,\ b_5 + a_5,\ b_6 + a_6,\ b_7 + a_7,\ b_8 + a_8,\ b_9 + a_9,\ b_{10} + a_{10},\ ...] \tag{43}$$

References

[1] Richard D. Jenks and Robert S. Sutor. *AXIOM.* Numerical Algorithms Group, Ltd., Oxford; Springer-Verlag, New York, 1992. The scientific computation system, With a foreword by David V. Chudnovsky and Gregory V. Chudnovsky.

[2] Weihua Liu. Relations between convolutions and transforms in operator-valued free probability. arXiv:1809.05789, 2018.